

ПЕРЕСТРАИВАЕМЫЙ ВЫСОКОЧАСТОТНЫЙ ГЕНЕРАТОР НА ОСНОВЕ СИНТЕЗАТОРА AD9850 И КОНТРОЛЛЕРА ARDUINO

Кулаков Владимир Геннадьевич

SPIN РИНЦ: 2111-7702

Контакт с автором: kulakovvlge@gmail.com

Цифровой синтезатор AD9850, выпускаемый корпорацией Analog Devices, способен формировать сигналы синусоидальной и прямоугольной формы с частотой от 0 до 62,5 МГц. Частота и фаза выходного сигнала AD9834 являются программируемыми: программирование осуществляется с внешнего микроконтроллера либо по параллельному, либо по последовательному интерфейсу.

На рисунке 1 в качестве примера показана схема генератора синусоидального сигнала, построенного на основе микросхемы AD9850BRSZ и контроллера Arduino. Частота сигнала на выходе генератора определяется информацией, поступающей с инкрементального энкодера на контроллер Arduino, программой, загруженной в этот контроллер, а также частотой тактового сигнала, подаваемого на микросхему DA1 синтезатора сигнала AD9850 с кварцевого генератора DD1. Напряжение питания +5В, необходимое для работы кварцевого генератора и синтезатора, обеспечивает стабилизатор, встроенный в контроллер Arduino.

В данном примере используется тактовый сигнал с частотой 80 МГц. Код K выходной частоты F для синтезатора AD9850 в таком случае вычисляется по следующей формуле:

$$K = F \times 2^{32} / 80000000.$$

Результат вычислений должен быть округлен до 32-разрядного целого числа без знака, а затем загружен в регистр синтезатора, хранящий код частоты.

Пример программы, позволяющей перестраивать частоту генератора от 1 до 20 МГц с шагом 50 кГц, приведен в листинге 1.




Рисунок 1. Схема высокочастотного генератора на микросхеме AD9850 с контроллером Arduino и буферным усилителем BUF634T на выходе

Листинг 1. Программа, предназначенная для управления работой цифрового синтезатора сигналов AD9850

```
#include <Arduino.h>
#define W_CLK      8 // AD9850: W_CLK
#define FQ_UD     9 // AD9850: FQ_UD
#define RESET    10 // AD9850: RESET
#define DATA    11 // AD9850: Serial data
#define pin_CLK  2 // Энкодер: пин А
#define pin_DT   4 // Энкодер: пин В
#define PosLim  380 // Ограничитель изменения положения энкодера

const float fgen = 80.0E6; // Частота сигнала ГТИ, Гц
const float sf = 1.0E6; // Начальное значение выходной частоты, Гц
const float df = 50000.0; // Шаг приращения частоты, Гц
volatile long Position = 0; // Текущее значение положения энкодера
long oldPosition = 0; // Предыдущее положение энкодера

float ff; // Код частоты в формате с плавающей точкой
unsigned long f; // Двоичный код выходной частоты
```

```

// Процедура для обработки прерывания от энкодера
void EncoderRotate() {
    if (digitalRead(pin_CLK) == digitalRead(pin_DT)) {
        if(Position < PosLim) Position++;
    } else {if(Position > 0) Position--;}
}

// Процедура инициализации AD9850
void AD9850init(){
    digitalWrite(RESET, HIGH);
    digitalWrite(RESET, LOW);
    digitalWrite(W_CLK, HIGH);
    digitalWrite(W_CLK, LOW);
    digitalWrite(FQ_UD, HIGH);
    digitalWrite(FQ_UD, LOW);
}

// Процедура для передачи кода частоты синтезатору AD9850
// по последовательному интерфейсу
void AD9850setfreq(unsigned long Freq){
    int i;
    unsigned long d;
    d = Freq;
    // Загрузка кода частоты
    for(i=0; i<32; i++)
    {
        if(d & 1 != 0) digitalWrite(DATA, HIGH);
        else digitalWrite(DATA, LOW);
        digitalWrite(W_CLK, HIGH);
        digitalWrite(W_CLK, LOW);
        d = d >> 1;
    }
    // Загрузка управляющего байта, заполненного нулями
    digitalWrite(DATA, LOW);
    for(i=0; i<8; i++)
    {
        digitalWrite(W_CLK, HIGH);
        digitalWrite(W_CLK, LOW);
    }
    // Завершение загрузки данных
    digitalWrite(FQ_UD, HIGH);
    digitalWrite(FQ_UD, LOW);
}

```

```

void setup() {
    int i;
    pinMode(W_CLK, OUTPUT);
    pinMode(FQ_UD, OUTPUT);
    pinMode(DATA, OUTPUT);
    pinMode(RESET, OUTPUT);
    pinMode(pin_CLK, INPUT);
    pinMode(pin_DT, INPUT);

    // Вычисляем начальный код частоты в формате с плавающей точкой
    ff = sf*(float)0x100001*(float)0x100001/fgen;
    // Получаем двоичный код выходной частоты
    f = (unsigned long)ff;
    // Инициализируем микросхему AD9850
    AD9850init();
    // Загружаем код начального значения частоты
    AD9850setfreq(f);

    // Подключаем обработчик прерывания энкодера
    attachInterrupt(digitalPinToInterrupt(pin_CLK),
        EncoderRotate, RISING);
}

void loop() {
    if (oldPosition != Position)
    {
        // Запоминаем текущую позицию
        oldPosition = Position;
        // Вычисляем код частоты в формате с плавающей точкой
        ff = (sf + df*(float)Position)*(float)0x100001*
            (float)0x100001/fgen;
        // Получаем двоичный код выходной частоты
        f = (unsigned long)ff;
        // Загружаем код частоты
        AD9850setfreq(f);
    }
}

```

При повороте ручки энкодера в ту или иную сторону генерируется прерывание и контроллер либо увеличивает, либо уменьшает значение переменной-счетчика положения энкодера `Position`. После любого изменения значения данной переменной контроллер заново вычисляет код частоты и загружает его в микросхему AD9850.

В диапазоне от 1 до 10 МГц амплитуда выходного сигнала на активной нагрузке с сопротивлением 50 Ом составляет 3,4 В. Форма сигнала на выходе генератора при частоте колебаний 1 МГц показана на рисунке 2.




Рисунок 2. Осциллограмма сигнала на выходе генератора при частоте 1 МГц

При увеличении частоты до 20 МГц амплитуда сигнала уменьшается до 2,2 В (рисунок 3).




Рисунок 3. Осциллограмма сигнала на выходе генератора при частоте 20 МГц

Теоретически, синтезатор AD9850 способен вырабатывать сигнал с частотой, составляющей до половины от частоты тактового сигнала, поступающего от кварцевого генератора. Однако в том случае, когда частота выходного сигнала синтезатора превышает четверть тактовой частоты, становится заметной

паразитная модуляция выходного сигнала. Таким образом, рабочий диапазон генератора может быть расширен до 30 МГц при использовании тактового генератора, вырабатывающего сигнал с более высокой частотой, например, 125 МГц.

Список использованной литературы

1. CMOS, 125 MHz Complete DDS Synthesizer AD9850, Rev. H – Analog Devices, Inc., 2004.
2. Low Cost, 300 MHz Voltage Feedback Amplifiers AD8055/AD8056, Rev. E – Analog Devices, Inc., 2001.
3. BUF634 250-mA High-Speed Buffer – Texas Instruments Incorporated, 2019.
4. Кулаков В.Г. Высокочастотный генератор на микросхеме AD9833 с многозвенным RC-фильтром. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/200820074529.pdf> (дата обращения: 20.08.2020).
5. Кулаков В.Г. Генератор на основе синтезатора AD9850, предназначенный для настройки фильтров с резонаторами. [Электронный ресурс]. URL: <http://new-idea.kulichki.net/pubfiles/210609181604.pdf> (дата обращения: 09.06.2021).

© В.Г. Кулаков, 2021